/**
 * Copyright (c) 2007-2008, the original author(s)
 * 
 * All rights reserved.
 * 
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 * 
 *     * Redistributions of source code must retain the above copyright notice,
 *       this list of conditions and the following disclaimer.
 *     * Redistributions in binary form must reproduce the above copyright
 *       notice, this list of conditions and the following disclaimer in the
 *       documentation and/or other materials provided with the distribution.
 *     * Neither the name of the Prana Framework nor the names of its contributors
 *       may be used to endorse or promote products derived from this software
 *       without specific prior written permission.
 * 
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
 * "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
 * LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
 * A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR
 * CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL,
 * EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO,
 * PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR
 * PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF
 * LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING
 * NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 */
package org.pranaframework.domain {
	
	import org.pranaframework.reflection.Constant;
	import org.pranaframework.reflection.Type;
	
	/**
	 * Base class for enumerations.
	 * 
	 * @author Christophe Herreman
	 */
	public class Enum extends Entity implements IEnum {
		
		/**
		 * Cached type info
		 */
		private static var _type:Type;
		
		private var _value:*;
		
		/**
		 * Creates a new Enum object.
		 * 
		 * @param id the id of the enum object
		 * @param value the value of the enum object
		 */
		public function Enum(id:*, value:*) {
			super(id);
			_value = value;
		}
		
		/**
		 * @inherited
		 */
		public function get value():* {
			return _value;
		}
		
		/**
		 * Sets the value for this enum entry.
		 */
		public function set value(v:*):* {
			_value = v;
		}
		
		/**
		 * Returns the same enum instead of cloning it.
		 */
		override public function clone():* {
			return this;
		}
		
		/**
		 * Looks up an enum entry for the given class by its id.
		 * 
		 * @param id the id of the enum entry to get
		 * @param clazz the class that defines the enumeration
		 */
		public static function fromID(id:*, clazz:Class):IEnum {
			var result:IEnum;
			var type:Type;
			
			if (_type) {
				type = (_type.clazz == clazz) ? _type : Type.forClass(clazz);			
			}
			else {
				type = Type.forClass(clazz);
			}
			
			 // cache type info
			_type = type;
			
			for (var i:int = 0; i<type.staticConstants.length; i++) {
				var constant:Constant = type.staticConstants[i];
				if (constant.type.clazz == clazz) {
					var enum:IEnum = constant.getValue();
					if (id == enum.id) {
						result = enum;
						break;
					}
				}
			}
			return result;
		}
		
	}
}